

12 Some efficiency concerns

Recall that we first defined the *reverse* function by recursion:

$$\begin{aligned} \text{reverse } [] &= [] \\ \text{reverse } (x : xs) &= \text{reverse } xs ++ [x] \\ &= \text{snoc } (\text{reverse } xs) x \\ &= \text{flip snoc } x (\text{reverse } xs) \end{aligned}$$

Deduce from this that

$$\text{reverse} = \text{fold } (\text{flip snoc}) [] \text{ where } \text{snoc } xs\ x = xs ++ [x]$$

However, this algorithm is quadratic: it takes about $\frac{1}{2}n^2$ steps to reverse a list of length n . Why is this? Each catenation

$$\begin{aligned} [] ++ ys &= ys \\ (x : xs) ++ ys &= x : (xs ++ ys) \end{aligned}$$

(or $(++ys) = \text{fold } (:) ys$) takes a number of steps linear in the length of its left argument. It follows that *snoc* takes a number of steps linear in its list argument, and *reverse* applies *snoc* to (the reverse of) each tail of its argument.

The insight is that we could accumulate the answer: invent

$$\text{revcat } ys\ xs = \text{reverse } xs ++ ys$$

Notice that this is intended as a specification, not the definition for execution: evaluating this would be at least as bad as the existing *reverse*. We could however use *revcat* to calculate

$$\begin{aligned} &\text{reverse } xs \\ = &\{ \text{unit of } (++) \text{ (proof?)} \} \\ &\text{reverse } xs ++ [] \\ = &\{ \text{specification of } \text{revcat} \} \\ &\text{revcat } []\ xs \end{aligned}$$

and then make the $(++)$ vanish, by synthesizing a $(++)$ -less recursive definition of *revcat*

$$\begin{aligned} &\text{revcat } ys\ [] \\ = &\{ \text{specification of } \text{revcat} \} \\ &\text{reverse } [] ++ ys \\ = &\{ \text{definition of } \text{reverse} \} \\ &[] ++ ys \\ = &\{ \text{definition of } (++) \} \\ &ys \end{aligned}$$

and for non-empty lists

$$\begin{aligned}
 & \text{revcat } ys \ (x : xs) \\
 = & \ \{ \text{specification of revcat} \} \\
 & \text{reverse } (x : xs) \ ++ \ ys \\
 = & \ \{ \text{definition of reverse} \} \\
 & (\text{reverse } xs \ ++ \ [x]) \ ++ \ ys \\
 = & \ \{ \text{associativity of } (++) \} \\
 & \text{reverse } xs \ ++ \ ([x] \ ++ \ ys) \\
 = & \ \{ \text{definition of } (++) \} \\
 & \text{reverse } xs \ ++ \ (x : ys) \\
 = & \ \{ \text{specification of revcat} \} \\
 & \text{revcat } (x : ys) \ xs
 \end{aligned}$$

This gives us a definition

```

> reverse = revcat []
>           where revcat ys    [] = ys
>                 revcat ys (x:xs) = revcat (x:ys) xs

```

This one is linear in the length of the list being reversed: each call of *revcat* corresponds to one of the conses in the list, and each call does a constant amount of work before the recursive call.

The correspondence between conses and calls of *revcat* suggests that we think of a fold, but it is not a *fold*. Compare it with

$$\begin{aligned}
 \text{tailfold } s \ n \ [] &= n \\
 \text{tailfold } s \ n \ (x : xs) &= \text{tailfold } s \ (s \ n \ x) \ xs
 \end{aligned}$$

and by inspection $\text{revcat } ys = \text{tailfold } (\text{flip } (:)) \ ys$ so

$$\text{reverse} = \text{tailfold } (\text{flip } (:)) \ []$$

12.1 Flattening trees

The flatten function for

```

> data BTree a = Leaf a | Fork (BTree a) (BTree a)

```

is $\text{flatten} :: BTree \ \alpha \rightarrow [\alpha]$ for which

$$\begin{aligned}
 \text{flatten } (\text{Leaf } x) &= [x] \\
 \text{flatten } (\text{Fork } ls \ rs) &= \text{flatten } ls \ ++ \ \text{flatten } rs
 \end{aligned}$$

The length of the result is the number of leaves in the tree, the size of the tree

$$size = foldBTree (const 1) (+)$$

however in general it takes more steps than that to produce it.

For a balanced tree of size n there will be a $(++)$ at the root that takes about $\frac{1}{2}n$ steps, below that two that take $\frac{1}{4}n$ steps each, and so on, which amounts to about $\frac{1}{2}n \log n$ steps. If the tree has a long left spine, the algorithm can be as bad as quadratic.

As before the insight is that we should specify

$$flatcat\ t\ ys = flatten\ t ++ ys$$

and synthesise

$$\begin{aligned} & flatcat\ (Leaf\ x)\ ys \\ = & \{ \text{specification of } flatcat \} \\ & flatten\ (Leaf\ x) ++ ys \\ = & \{ \text{definition of } flatten \} \\ & [x] ++ ys \\ = & \{ \text{definition of } (++) \} \\ & x : ys \end{aligned}$$

and

$$\begin{aligned} & flatcat\ (Fork\ ls\ rs)\ ys \\ = & \{ \text{specification of } flatcat \} \\ & flatten\ (Fork\ ls\ rs) ++ ys \\ = & \{ \text{definition of } flatten \} \\ & (flatten\ ls ++ flatten\ rs) ++ ys \\ = & \{ \text{associativity of } (++) \} \\ & flatten\ ls ++ (flatten\ rs ++ ys) \\ = & \{ \text{specification of } flatcat \} \\ & flatcat\ ls\ (flatcat\ rs\ ys) \end{aligned}$$

so $flatcat = foldBTree\ (:) (\cdot)$ and $flatten\ t = foldBTree\ (:) (\cdot)\ t\ []$ and, relying on the associativity of $(++)$, synthesis has produced a linear algorithm from a less efficient one.

12.2 Associativity and folds

When is $fold\ (\oplus)\ e = tailfold\ (\otimes)\ f$?

Suppose we try to prove this by induction. It is chain complete, and both sides are strict. Applying both sides to $[]$ shows that it is necessary that $e = f$. The substantial part of the proof is

$$\begin{aligned}
& \text{fold } (\oplus) e (x : xs) \\
= & \{ \text{definition of fold} \} \\
& x \oplus \text{fold } (\oplus) e xs \\
= & \{ \text{lemma to be proved} \} \\
& \text{tailfold } (\otimes) (e \otimes x) xs \\
= & \{ \text{definition of tailfold} \} \\
& \text{tailfold } (\otimes) e (x : xs)
\end{aligned}$$

The essence of the result is the missing lemma, again to be proved by induction.

The assertion to be proved is chain complete. If $xs = \perp$ conclude that $x \oplus \perp = \perp$ for all x , so (\oplus) must be strict in its second argument. If $xs = []$ conclude that $e \otimes x = x \oplus e$. The substantial part of the proof of the lemma is

$$\begin{aligned}
& \text{tailfold } (\otimes) (e \otimes x) (y : ys) \\
= & \{ \text{definition of tailfold} \} \\
& \text{tailfold } (\otimes) ((e \otimes x) \otimes y) ys \\
= & \{ \text{suppose } (a \otimes b) \otimes c = a \otimes (b \odot c) \} \\
& \text{tailfold } (\otimes) (e \otimes (x \odot y)) ys \\
= & \{ \text{induction hypothesis} \} \\
& (x \odot y) \oplus \text{fold } (\oplus) e ys \\
= & \{ \text{suppose } (a \odot b) \oplus c = a \oplus (b \oplus c) \} \\
& x \oplus (y \oplus \text{fold } (\oplus) e ys) \\
= & \{ \text{definition of fold} \} \\
& x \oplus \text{fold } (\oplus) e (y : ys)
\end{aligned}$$

Notice that this is a proof for all values of x , and the induction hypothesis is that it holds for all x and a specific ys .

Collecting the requirements:

$$\text{fold } (\oplus) e = \text{tailfold } (\otimes) e$$

is proved for right-strict (\oplus) , provided $e \otimes x = x \oplus e$ and provided there is a (\odot) for which $a \otimes (b \odot c) = (a \otimes b) \otimes c$ and $(a \odot b) \oplus c = a \oplus (b \oplus c)$.

The obvious case is when all three of (\oplus) , (\otimes) and (\odot) are equal, are right-strict,

are associative, and have e as a left and right unit.

$$\begin{aligned}
 \text{sum} &= \text{fold } (+) 0 \\
 &= \text{tailfold } (+) 0 \\
 \text{product} &= \text{fold } (\times) 1 \\
 &= \text{tailfold } (\times) 1 \\
 \text{concat} &= \text{fold } (++) [] \\
 &\neq \text{tailfold } (++) []
 \end{aligned}$$

this inequality is because $xs ++ \perp \neq \perp$. The *fold* form produces output when applied to an infinite list of lists provided enough of them are non-empty, but the *tailfold* form cannot produce any output for an infinite (or partial) input.

12.3 Bounding space

One reason for preferring *tailfold* $(+) 0$ to *fold* $(+) 0$ is that the *fold* is generally obliged to build up the whole expression before any evaluation:

$$\begin{aligned}
 &\text{fold } (+) 0 [1, 2, 3, 4] \\
 = &1 + \text{fold } (+) 0 [2, 3, 4] \\
 = &1 + (2 + \text{fold } (+) 0 [3, 4]) \\
 = &1 + (2 + (3 + \text{fold } (+) 0 [4])) \\
 = &1 + (2 + (3 + (4 + \text{fold } (+) 0 []))) \\
 = &1 + (2 + (3 + (4 + 0))) \\
 = &1 + (2 + (3 + 4)) \\
 = &1 + (2 + 7) \\
 = &1 + 9 \\
 = &10
 \end{aligned}$$

whereas the *tailfold* can be seen as evaluating the expression as it goes. In fact, because of lazy evaluation

$$\begin{aligned}
 &\text{tailfold } (+) 0 [1, 2, 3, 4] \\
 = &\text{tailfold } (+) (0 + 1) [2, 3, 4] \\
 = &\text{tailfold } (+) ((0 + 1) + 2) [3, 4] \\
 = &\text{tailfold } (+) (((0 + 1) + 2) + 3) [4] \\
 = &\text{tailfold } (+) ((((0 + 1) + 2) + 3) + 4) [] \\
 = &(((0 + 1) + 2) + 3) + 4 \\
 = &((1 + 2) + 3) + 4 \\
 = &(3 + 3) + 4 \\
 = &6 + 4 \\
 = &10
 \end{aligned}$$

the same space build-up can happen.

To prevent it, *tailfold* would have to be made strict in this argument.

```
> tailfold' s n [] = n
> tailfold' s n (x:xs) = let !snx = s n x in tailfold' s snx xs
```

The `!` decoration ensures that the variable *snx* is evaluated before the recursive call.

12.4 Fast exponentiation

On the face of it, calculating x^n appears to require about n multiplication. But multiplication is associative, so $x^{2^n} = (x^2)^n$ and x^{2^n} can be calculated in only one more multiplication than x^n . So we could specify *pow x n = xⁿ* and synthesize

```
pow x 0 = 1
pow x n | even n = pow (x*x) (n `div` 2)
          | odd n  = pow x (n-1) * x
```

This function will be called no more than $2 \log n$ times in x^n .

However, just like the *fold* version of *product*, this function will unnecessarily build up a big expression before any evaluation. Specify *power y x n = pow x n × y* and synthesize

$$\begin{aligned} \text{power } y \text{ } x \text{ } 0 &= \text{pow } x \text{ } 0 \times y \\ &= 1 \times y \\ &= y \\ \text{power } y \text{ } x \text{ } n \mid \text{even } n &= \text{pow } x \text{ } n \times y \\ &= \text{pow } (x \times x) (n \text{ div } 2) \times y \\ &= \text{power } y \text{ } (x \times x) (n \text{ div } 2) \\ \text{power } y \text{ } x \text{ } n \mid \text{odd } n &= \text{pow } x \text{ } n \times y \\ &= (\text{pow } x \text{ } (n - 1) \times x) \times y \\ &= \text{pow } x \text{ } (n - 1) \times (x \times y) \\ &= \text{power } (x \times y) \text{ } x \text{ } (n - 1) \end{aligned}$$

We could also abstract on the multiplication:

```
> power (*) y x n == x^n*y
>      | n == 0 = y
>      | even n = power (*) y (x*x) (n `div` 2)
>      | odd n  = power (*) (x*y) x (n-1)
```

Notice that the development of this code used only the associativity of (\times) , so it will calculate other repeated operations such as repeated matrix multiplication.