

14 Countdown, dynamic programming solutions

The previous lecture defined solvers for the Countdown problem which took the form

$$\begin{aligned} \text{solutions } ns \ n = & [(e, v) \mid \text{ys} \leftarrow \text{subs } ns, \\ & \text{cs} \leftarrow \text{permutations } \text{ys}, \\ & (e, v) \leftarrow \text{results } \text{cs}, \\ & v == n] \end{aligned}$$

and where the *results* function was repeatedly refined. In each case the *results* function then decomposes *cs* into subsequences on which it builds expressions. Consequently the same work is being done repeatedly.

14.1 Tabulation for efficiency

The well-known Fibonacci sequence, the image of the natural numbers under

```
> fib :: Int -> Integer
> fib 0 = 0
> fib 1 = 1
> fib n = fib (n-1) + fib (n-2)
```

is most naturally specified by that function definition. However, this is an inefficient way of computing elements of the sequence.

Since $fib\ n \approx \frac{1}{\sqrt{5}} \left(\frac{1+\sqrt{5}}{2} \right)^n$, and since the result is produced by adding 1 to itself (and to some zeroes) this function (read as a program) gives an exponentially slow way of calculating elements of the sequence.

Observe that in calculating *fib n* there are about $2 \times fib\ n$ calls of *fib* but only *n* different calls, since each call has an argument less than *n*. It would be better to calculate each of these once, and then use their values as needed.

One way of understanding tabulation is first to extract the *kernel* of the recursion for *fib*.

```
> fibK :: (Int -> Integer) -> Int -> Integer
> fibK f 0 = 0
> fibK f 1 = 1
> fibK f n = f (n-1) + f (n-2)
```

This function, if applied to *fib*, produces the same result, that is $fib = fib_K\ fib$, or $fib = fix\ fib_K$ where $fix\ f = f(fix\ f)$. Of course, $fix\ fib_K$ will not only implement the same function as *fib*, but also it calls *fib_K* exactly as many times as the original program called *fib*, and so has the same exponentially poor performance.

The *dynamic programming* strategy for evaluating a function such as *fib* is to construct a table of values of the function, and to implement the function by looking up entries in that table. The table itself is constructed by calling the kernel, but with the table-lookup implementation as an argument.

In the case of *fib*, which is a function from natural numbers, the table, *tab*, might be a list of values of the function, in order. Then $fib\ n = tab\ !!\ n$. The table can be constructed by mapping an implementation of *fib* over the domain of the function, $[0..]$. That implementation consists of the kernel of the recursion, but with recursive calls implemented by table lookup.

```
> tabulate :: ((Int -> a) -> (Int -> a)) -> (Int -> a)
> tabulate kernel = (tab 'get')
>                 where get = (!! )
>                 tab = map (kernel (tab 'get')) dom
>                 dom = [0..]
```

With this definition, $tabulate\ fib_K = fib$ is an implementation of *fib* which only ever applies the body of fib_K with second argument *n* once for each value of *n*.

The table *tab* is infinite, but only a finite amount of it (the first $n + 1$ elements) will be explored and filled in for any call of $tabulate\ fib_K\ n$. However, that piece of the table which is instantiated grows with *n* so the cost of reducing the time taken by the program is an increase in the space which it must occupy.

(This is not a linear implementation of *fib*, because it calls *get* about twice for each natural number less than *n*, and *get* is linear in its numerical argument, so the resulting implementation of *fib* is quadratic.)

Strictly speaking, the name *dynamic programming* is usually used to mean tabulation, with predetermined bounds on the table to be used, along with a schedule for evaluating the elements of the table. The schedule guarantees that each entry in the table is filled in before it is looked up. In our program, we rely on lazy evaluation to evaluate the table entries in an order which provides the answers when they are needed. This works provided that there is an ordering on the arguments with respect to which recursive calls are smaller than the call which causes them. Such an ordering must have existed in order for the original recursive function to terminate. In the case of *fib*, this is just the usual ordering on numbers.

14.2 Tabulating results

The calculation of solutions to a countdown problem can reorganised to use a function which lists the solution expressions that use all of the numbers in some list. (The reasons for wanting to do that will become clear later.)

```
> solutions :: ([Int] -> [Result]) -> [Int] -> Int -> [Expr]
> solutions results ns n = [ e | (e,v) <- results ns, v == n ]
```

```
> results :: [Int] -> [Result]
> results ns = [ ev | ls <- subs ns, ev <- fix resultsK ls ]
```

The *results* function lists those expressions that use exactly *ls* for each subsequence *ls* of its argument *ns*, using a recursion described by *resultsK*.

```
> resultsK :: ([Int]-> [Result]) -> ([Int] -> [Result])
> resultsK f [] = []
> resultsK f [n] = [ (Val n, n) | n > 0 ]
> resultsK f ns = [ (App o l r, apply o x y)
>                   | (ls, rs) <- subsplits ns,
>                   (l,x) <- f ls,
>                   (r,y) <- f rs,
>                   o <- ops,
>                   needed o l r,
>                   useful o x y ]
```

This recursion filters for *useful* arrangements of values of subexpressions, but also for the associativity properties eliminated by the less efficient implementation *generatedresults* from the previous lecture.

Binary expressions are constructed by dividing *ns* into two subsequences, rather than just an initial segment and a final segment. The *subsplits* function divides a non-empty list into a pair of non-empty subsequences in all possible ways. (They must be non-empty sequences, because the program later relies on *the other* element of the pair being smaller than the list being split.)

```
> subsplits :: [a] -> [[a], [a]]
> subsplits (x:xs)
>   | null xs    = []
>   | otherwise = [ ([x],xs), (xs,[x]) ] ++
>                 concat [ [ (x:ls,rs), (ls,x:rs) ]
>                           | (ls,rs) <- subsplits xs ]
```

This leads to fewer arguments to *results* than the strategy in the previous lecture, because the numbers in the argument of *resultsK* are always in the order in which they appeared in the problem.

There is as yet no tabulation in this new implementation, but it is slightly faster than the fastest solution from the previous lecture.

```
Lecture13> head (approx usefulresults numbers 831)
(7+(3*((1+10)*25)),832)
(1.81 secs, 845,372,312 bytes)
```

```
Lecture14> head (approx results numbers 831)
(7+(3*((1+10)*25)),832)
(1.23 secs, 509,608,648 bytes)
```

14.3 Naïve tabulation

A table for the *results* function has to represent a function $[Int] \rightarrow Int$, with domain *subs ns* for some six numbers *ns*.

One perfectly general solution represents an $Eq\ a \Rightarrow a \rightarrow b$ by an *association list* of pairs $[(a, b)]$, which for clarity later, we tag with the constructor *Map*

```
> data Map a b = Map [(a,b)]
```

then the association list with domain *xs* is just a list of pairs of elements of the domain together with their image under the function which the list represents.

```
> toMap :: (a -> b) -> [a] -> Map a b
> toMap f xs = Map [ (x, f x) | x <- xs ]
```

Values of $f\ x$ can be recovered from the map by finding the (or the first) value paired with *x*.

```
> getMap :: Eq a => Map a b -> a -> b
> Map m 'getMap' x = head [ b | (a,b) <- m, a == x ]
```

With this implementation of a mapping the fixed point of *results_K* can be tabulated with an association list indexed by the subsequences of *ns*.

```
> tabresults :: [Int] -> [Result]
> tabresults ns = [ ev | ls <- dom, ev <- tab 'getMap' ls ]
>               where tab = toMap (resultsK (tab 'getMap')) dom
>               dom = subs ns
```

This function constructs the table once only for each problem (one with numbers *ns*) so calls *results_K* only $2^6 = 64$ times.

```
Lecture14> head (approx tabresults numbers 831)
(7+(3*((1+10)*25)),832)
(0.57 secs, 209,261,672 bytes)
```

However it makes many calls of *getMap*, each of which involves exploring the table, which is itself 64 long.

14.4 Using a Trie for tabulation

The domain of the mapping is a set of lists, and a possibly more efficient strategy for tabulating a prefix-closed set of lists is to use a *Trie* (possibly meant to be pronounced *tree*) which is a particular form of *rose tree*.

```
> data Trie a b = Trie b (Map a (Trie a b))
```

The nodes of a *Trie* $a\ b$ for some function $f :: [a] \rightarrow b$ correspond to a set of prefixes of some lists from $[a]$ together with the values of the function at those prefixes. The root contains $f []$, and $f(x : xs)$ appears in the trie found by looking for x in the mapping in the root:

```
> getTrie :: Eq a => Trie a b -> [a] -> b
> (Trie y _) 'getTrie' [] = y
> (Trie _ m) 'getTrie' (x:xs) = (m 'getMap' x) 'getTrie' xs
```

How to construct a trie for a function on the subsequences of a list? One way is to consider the subsequences generated in lexicographical order.

```
> subs :: [a] -> [[a]]
> subs xs = [[]] ++ [ y:zs | (y:ys) <- tails xs, zs <- subs ys ]
```

Here the *tails* are the non-empty tails of a list.

```
> tails :: [a] -> [[a]]
> tails [] = []
> tails xs = xs : tails (tail xs)
```

An analogous structure produces a trie for the subsequences of xs .

```
> toTrie :: ([a] -> b) -> [a] -> Trie a b
> toTrie f xs = Trie (f []) (Map [ (y, toTrie (f.(y:)) ys)
>                                | y:ys <- tails xs ])
```

Here the function $f \cdot (y :)$ is tabulated by the trie that is reached from the root by looking up y , so a value found by looking up a path ys in that subtrie can be found by looking up $y : ys$ from the root.

The tabulation of the results looks very similar to the earlier one, except that *toTrie* produces a trie with a domain that is not its list argument, but the subsequences of that argument

```
> trieresults :: [Int] -> [Result]
> trieresults ns = [ ev | ls <- subs ns, ev <- tab 'getTrie' ls ]
>               where tab = toTrie (resultsK (tab 'getTrie')) ns
```

and the resulting program is significantly faster

```
Lecture14> head (approx trieresults numbers 831)
(7+(3*((1+10)*25)),832)
(0.37 secs, 157,684,760 bytes)
```

The effect is more noticeable with bigger problems, for example with seven numbers a trie is a clear factor of two better than an association list.